Sage scripts for the unit index question

Here we provide short scripts to check the coincidence between the factors in the class number formula for biquadratic fields $\mathbb{Q}(\sqrt{-m},\zeta_3)$ and the index of the reduction morphism $\mathcal{O}_{-m}[\zeta_3]^\times \to \mathcal{O}_m[\zeta_3]/(\zeta_3-1)^\times \cong \mathcal{O}_{-m}/(3)^\times$.

The following short SAGE-script goes through all square-free $m \leq 1000$ with $m \equiv 2 \bmod 3$, computes the class numbers of the fields $K = \mathbb{Q}(\sqrt{-m})$, $L = \mathbb{Q}(\sqrt{-m},\zeta_3)$ and $T = \mathbb{Q}(\sqrt{3m})$ and the factor $\frac{h_{\mathbb{Q}(\sqrt{-m},\sqrt{-3})}}{h_{\mathbb{Q}(\sqrt{-m})}h_{\mathbb{Q}(\sqrt{3m})}}$. Then it computes a fundamental unit and checks if the unit is ± 1 modulo the ideal (ζ_3-1). The final line checks that the reduction of the fundamental unit is ± 1 if and only if the class number formula factor is $\frac{1}{2}$. For more discussion and a table of the first few outputs, see Example 6.3 of the paper "ON FARRELL–TATE COHOMOLOGY OF GL3 OVER RINGS OF QUADRATIC INTEGERS" by BUI ANH TUAN, ALEXANDER D. RAHM AND MATTHIAS WENDT.

```
for m in range(1000):
    if mod(m,3) == 2 and Integer(m).is_squarefree():
       print('m='+str(m))
       K.<sqnm> = NumberField(x^2+m)
       R.<y> = PolynomialRing(K)
       L.<zeta_3> = K.extension(y^2+y+1)
       S.<c> = PolynomialRing(QQ)
       T.<b> = NumberField(c^2-3*m)
#       print('class number='+str(L.class_number()))
       factor = L.class_number()/(K.class_number()*T.class_number())
#       print('class number formula factor='+str(factor))
       fu = L.units()[0]
       ideal = L.ideal(zeta_3-1)
       if not ((fu - 1 in ideal) or (fu + 1 in ideal)) == (factor == 1/2):
          print('conjecture false')
```

For the case $m \equiv 1 \bmod 3$, only few modifications have to be made. Obviously the change to `mod(m,3) == 1` in the second line, and then the last line of the previous script has to be replaced by

```
       index = mod(ideal.ideallog(fu)[0],4)
       print('unit index='+str(index))
       if mod(index,2) == 1:
          print('exceptional case')
       if not ((index == 2) == (factor == 1)):
          print('conjecture false')
```

Here, the function `ideallog(fu)` is used to compute the index of the reduction of the fundamental unit in $\mathcal{O}_{-m}/(3)^\times \cong \mathbb{Z}/8\mathbb{Z}$. According to our expectation in Example 6.5 of the paper "ON FARRELL–TATE COHOMOLOGY OF GL3 OVER RINGS OF QUADRATIC INTEGERS" by BUI ANH TUAN, ALEXANDER D. RAHM AND MATTHIAS WENDT, the reduction of the fundamental unit should only be a generator in the case $m = 1$, otherwise it should have reduction equal to ± 1 (and thus yield a number of 5 orbits) if and only the class number formula factor is $\frac{1}{2}$. For more discussion and a table of outcomes, cf. the quoted Example 6.5.

Bui Anh Tuan, Faculty of Mathematics and Computer Science, University of Science, VNU-HCM, 227 Nguyen Van Cu Str., Dist. 5, Ho Chi Minh City, Vietnam
Email address: batuan@hcmus.edu.vn

Alexander D. Rahm, Laboratoire de mathématiques GAATI, Université de la Polynésie française, BP 6570, 98702 Faaa, French Polynesia
Email address: Alexander.Rahm@upf.pf

Matthias Wendt, Fachgruppe Mathematik/Informatik, Bergische Universität Wuppertal, Gaussstrasse 20, 42119 Wuppertal, Germany
Email address: wendt@math.uni-wuppertal.de